Canadian Technical Report of
Fisheries and Aquatic Sciences 1134

November 1982

LETCUR, THE LETHALITY CURVE PROGRAM

by

V. Zitko

Fisheries and Environmental Sciences
Department of Fisheries and Oceans
Biological Station
St. Andrews, N.B. EOG 2XO

This is the one hundred and fifty-third Technical Report from
the Biological Station, St. Andrews, N.B.

Cat. No. Fs 97-6/1134 ISSN 0706-6457

Correct citation for this publication:

Zitko, V. 1982. LETCUR, the lethality curve program. Can. Tech. Rep. Fish. Aquat. Sci. 1134, iii + 10 p.

ABSTRACT

Zitko, V. 1982. LETCUR, the lethality curve program. Can. Tech. Rep. Fish. Aquat. Sci. 1134, iii + 10 p.

A BASIC program is described to express LT50 (time to 50% mortality) as a three-parameter (K2, A, B) function of toxicant concentration (CW): LT50 = -ln (A-B/CW)/K2. A calculator program, published previously, is documented to facilitate its portability.

Key words: aquatic toxicology, acute toxicity, computer program

RÉSUMÉ

Zitko, V. 1982. LETCUR, the lethality curve program. Can. Tech. Rep. Fish. Aquat. Sci. 1134, iii + 10 p.

Le présent rapport contient la description d'un programme BASIC permettant d'exprimer le TL50 (temps requis pour 50% de mortalité) comme fonction à trois paramètres (K2, A, B) de la concentration du toxique (CW) : TL50 = -ln (A-B/CW)/K2. Une référence à un programme machine déjà publié en facilitera la portabilité.

INTRODUCTION

Lethality curves, graphs of time to LT50 plotted vs CW, are used widely in aquatic toxicology. They present graphically the "dose-mortality" relationship and are extrapolated frequently to estimate the incipient lethal level (lethality threshold).

The function

$$LT50 = -\ln (A-B/CW)/K2$$

has been suggested for the lethality curve (Zitko 1979). This suggestion was based on the one-compartment model of uptake of chemicals.

Expressing the lethality curve in this manner has several advantages over the usual plotting and storage of the graphs. The data are condensed into three parameters and may be stored readily in a data base. Comparison of the values of the parameters for a number of chemicals may lead to the development of structure/lethality relationships. Such comparisons are practically impossible or are at best only qualitative, when based on a visual comparison of graphs. In addition, incipient lethal levels are given consistently by the ratio A/B. Finding the values of the three parameters is straightforward but time-consuming, since it involves searching for a value of K2 which yields correlation coefficient $R \leq -0.9$ between EXP(-K2*LT50) and 1/CW. For each selected value of K2, linear regression must be calculated to obtain the corresponding values of the parameters A and B.

The original paper (Zitko 1979) contains a program to perform the calculations on the HP29C programmable calculator. The program has not been documented and, since the HP29C is no longer a common calculator, the transfer of the program to other calculators presents problems. A detailed documentation is given in this report to make the program easy to implement on any programmable calculator.

The main part of this report contains documentation of the program LETCUR, written in HP Terminal BASIC. The program calculates values of the three parameters and plots the lethality curve as well. The parameter calculation part is readily transferable to any computer with a reasonably "standard" BASIC interpreter. On the other hand, the plotting part may require considerable rewriting, although the algorithms, such as those for plotting logarithmic scales and data scaling, are generally applicable.

LETCUR also contains a "convenience" routine for the calculation of mean concentration of the toxicant if, as is the case in most static experiments, the concentration decreases exponentially with time. The routine also provides for the usual practice of changing the toxicant solution at 48 h of exposure.

PROGRAM LETCUR

In the attached listing, the calculation of the parameters K2, A, and B is performed by statements 100-765 and 3000-3150, the latter containing the linear regression subroutine. A major part of the listing (statements 770-1380) deals with plotting the lethality curve. Statements 4500-5010 are used optionally to calculate the mean concentration of the toxicant and to display the concentration vs time curve.

The arrays dimensioned in statement 100 are used as follows: C(20) toxicant concentrations, T(20) LT50's, G(20) calculated LT50's, Z(20) scaled logarithms of toxicant concentrations, D(20) logarithms of LT50's. The arrays may be expanded, but there are seldom more than 20 points in lethality tests.

Concentration units should be chosen so that the values are >1. This is because of the plotting format used. For the same reason, experimental data must be entered in decreasing order of concentration. In autoscaling (statements 845-960) the program assumes that the first and last toxicant concentration entered are the highest and the lowest, respectively.

The search for the optimum value of K2 is performed in three stages: First (statements 205-300) a preliminary value is found, which gives a correlation coefficient $R \leq -0.9$. Then (statements 320-435) the interval containing the optimum value is obtained. Finally (statements 460-640) the interval is narrowed to within 1% of the optimum K2 value. The whole algorithm is a generally applicable "one-dimensional" optimization. A convenient starting value of K2 is usually K2 = 1, but there is the option for the user to start with any value (statement 210). On completion of the optimization, values of the parameters K2, A and B, table of toxicant concentrations, experimental and calculated LT50's, and the incipient lethal level are printed (statements 645-765). The linear regression used during the optimization (statements 3000-3150) uses EXP(-K2*LT50) as the dependent and 1/CW as the independent variable. The calculations are routine otherwise.

In preparation for plotting, the program first determines whether the incipient lethal level is too close to the Y-axis and, if it is, the X-axis will be scaled to shift the incipient lethal level to the right. The variable G (statements 774 and 776) is the flag. PRINT CHR$(141) means "clear screen" (statement 800). Logarithms of experimental LT50's and toxicant concentrations are calculated. At the same time, the former are shifted to start from 0 (log 1; statements 810-840, see also 960). Next, the numbers of logarithmic cycles needed on the axes (M on the X-axis, N on the Y-axis) are determined (statements 845-960). The loops in statements 970-1210 draw the "ticks" on the axes. This is a generally useful algorithm for plots in logarithmic coordinates.

Printing labels and scales (statements 1212-1241) completes the preparation of the graph format. The flag Q9 indicates when the shortest LT50 is less than 10 h, and the Y-axis is shifted accordingly. Experimental points are plotted by statements 1232-1260. Following this, the incipient lethal level line is drawn and labeled (statements 1266-1296). Finally, the calculated lethality curve is plotted (statements 1298-1320) and the graph is labeled by the name of the compound, incipient lethal level, and the equation of the lethality curve (statements 1321-1328).

The described automatic graph formatting and plotting routine worked well in all of our applications. However, such automated routines tend to run into problems in some unusual or unanticipated applications. Statements 4000-4030 mean "assign source to graphics", "assign destination to plotter (HP-IB#6)", and "transfer file", respectively. Their execution results in copying the graphics display on the plotter. The listing is concluded by the optional routine calculating mean concentration of the toxicant. The calculation assumes that the concentration decreases exponentially with time. To check the validity of this assumption, the concentrations are plotted against time for a visual check. It is also possible to use only selected points for the calculation. The parameters "slope" (M) and "intercept" (C) are determined by linear regression on the linearized function ln(CW) = MT+C. The concentrations should be entered relative to concentration at time t = 0 (statements 4500-4820). The subroutines starting in statements 4920 and 4980 draw the points and the calculated curve, respectively.

Statements 4840-4890 calculate the mean concentration. This depends on the length of exposure, that is on the LT50 at a given concentration and is calculated from the formula:

$$CW_{av} = 1/LT50 * Co * \int_0^{LT50} EXP(-M*t)*dt$$

where CW_{av} = mean concentration,
Co = nominal concentration multiplied by intercept,
M = slope.

The calculation assumes that the toxicant solution is changed at 48 h. Accordingly, for LT50 $<$ 48 h the calculation proceeds as outlined, but for LT50 $>$ 48 h the concentration is averaged to 48 h (statement 4872), another mean is calculated for the exposure time in the new toxicant solution (statement 4873) and these two means are averaged again (statement 4874). If the experiment was carried out for longer than 48 h without solution change then statement 4865 must be modified. A modification would also be required for solution changes taking place at shorter time intervals.

After completion of these calculations, the program proceeds to fit the lethality curve (statement 4910). There is no provision for sending the CW vs time graph to the plotter. If desired, this operation may be easily accomplished manually.

LETCUR ON HP29C

The documentation below is largely self-explanatory. The calculation proceeds in much the same way as in LETCUR except that the K2 search is done manually and the optimization is left completely to the judgment of the user. Also, mean toxicant concentrations are not calculated and must be entered by the user. The program consists of five subroutines (called labels, LBL). Some are used internally by the program, others are selected by the user. LBL0 is used to enter data. The parameter search is performed by LBL1 which calls LBL2 and LBL3. Values of LT50 for given CW and fitted parameters K2, A, and B are calculated by LBL9. Great care must be exercised in proper calculator initialization for the various operations.

ACKNOWLEDGMENTS

I thank Ms. R. Garnett for editorial assistance, Mrs. J. Hurley and Mrs. B. Fawkes for typing the manuscript and arranging the report, Ms. Madelyn M. Irwin, and Ms. R. Garnett for patient proofreading. Drs. D. W. McLeese and S. Ray commented on the manuscript.

REFERENCES

Zitko, V. 1979. An equation of lethality curves in tests with aquatic fauna. Chemosphere 8: 47-51.

```
&JD
10 PRINT "LETCUR REV.2, APRIL 11,1980"
15 PLOTR
20 INPUT "COMPOUND NAME (MAX.18 CHARACTERS) ",T$
100 DIM C(20),T(20),E(20),G(20),D(20),Z(20)
110 INPUT "NUMBER OF POINTS (MAX.20) ",Q
112 PRINT "CHOOSE CONCENTRATION UNITS SO THAT VALUES ARE >1"
115 INPUT "CONCENTRATION UNITS (MG/L,UG/L),TYPE UNITS ",C$
116 PRINT "A SUBROUTINE IS AVAILABLE FOR CALCULATION OF AVERAGE CONCENTRATIONS"
117 INPUT "SUBROUTINE: YES(1),NO(2) ",Q7
118 IF Q7=1 THEN 4500
120 PRINT "INPUT CONCENTRATION AND LT50 IN DECREASING ORDER OF CONCENTRATIONS"
140 FOR I=1 TO Q
150 PRINT "POINT";I
160 INPUT "CONCN. ",C(I)
185 INPUT "LT50 ",T(I)
197 NEXT I
205 PRINT "K2 ESTIMATION BEGINS"
210 INPUT "K2 (RECOMMENDED K2=1) ",K2
220 FOR I=1 TO 20
230 K2=.5*K2
240 GOSUB 3000
250 IF R<-.9 THEN 310
260 NEXT I
270 PRINT "FAILED TO CONVERGE"
275 PRINT "FINAL K2=";K2
280 INPUT "REPEAT (1),END(2) ",Q1
290 IF Q1=1 THEN 205
300 GOTO 1330
310 PRINT "PRELIMINARY K2 ESTIMATE",K2
315 PRINT "------------------------------"
320 PRINT "K2 OPTIMIZATION BEGINS"
330 B2=B
340 V2=K2
350 B1=B
360 V1=K2
370 K2=.5*K2
380 GOSUB 3000
390 IF B<B1 THEN 350
400 B1=B
410 V1=K2
415 PRINT "-------------------------------------"
420 PRINT "INITIAL K2 RANGE",V1,V2
430 PRINT "INITIAL B RANGE",B1,B2
435 PRINT "-------------------------------"
460 V3=.5*(V1+V2)
470 K2=V3
480 GOSUB 3000
490 B3=B
500 K2=.5*(V2+V3)
510 GOSUB 3000
520 IF B<B3 THEN 600
530 K2=.5*(V1+V3)
540 GOSUB 3000
550 V2=V3
560 V3=K2
570 B2=B3
580 B3=B
590 GOTO 640
600 V1=V3
610 V3=K2
620 B1=B3
630 B3=B
640 IF (V2-V1)/V3>.01 THEN 460
645 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$"
646 PRINT "CONVERGED. FINAL VALUES:"
647 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$"
650 PRINT "K2=";V3,"A=";A,"B=";B3
670 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$"
675 PRINT
680 PRINT "                              LT50"
690 PRINT " CONCENTRATION  -----------------------"
700 PRINT C$;"                    MEASURED      CALCULATED"
710 PRINT "=============================================="
720 FOR I=1 TO Q
725 IF A+B/C(I)<=0 THEN 755
730 G(I)=-LOG(A+B/C(I))/K2
740 PRINT C(I),T(I),G(I)
```

```
750 NEXT I
751 GOTO 760
755 PRINT C(I),T(I),"OUT OF RANGE"
760 PRINT "========================================"
761 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$"
765 PRINT "THRESHOLD IS",-B/A
766 PRINT "$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$"
770 INPUT "GRAPH:YES(1),NO(2) ",Q2
772 REM TO SHIFT IF THRESHOLD LINE COMES CLOSE TO THE Y AXIS
773 IF .434295*LOG(-B/A)<.3 THEN 776
774 G=0
775 GOTO 780
776 G=1
780 IF Q2=1 THEN 800
790 GOTO 1330
800 PRINT CHR$(141)
810 FOR I=1 TO Q
820 Z(I)=.434295*LOG(C(I))-INT(.434295*LOG(C(Q)))
830 D(I)=.434295*LOG(T(I))
840 NEXT I
845 REM DETERMINATION OF SCALING FACTORS
850 R=C(1)/C(Q)
860 R1=.434295*LOG(R)
870 R2=INT(R1)
880 M=R2+2+G
890 R3=T(Q)/T(1)
900 R4=.434295*LOG(R3)
910 R5=INT(R4)
920 N=R5+2
940 LOCATE (40,190,7,97)
950 FRAME
960 SCALE (0,M,0,N)
965 REM DRAWING OF SCALES
970 FOR J=0 TO M
980 FOR I=1 TO 10
990 MOVE (.434295*LOG(I)+J,.05)
1000 IF I=10 THEN 1030
1010 DRAW (.434295*LOG(I)+J,0)
1020 GOTO 1070
1030 FOR K=0 TO N STEP .05
1040 PLOT (.434295*LOG(I)+J,K)
1050 PENUP
1060 NEXT K
1070 MOVE (.434295*LOG(I)+J,N)
1080 DRAW (.434295*LOG(I)+J,N-.05)
1090 IF J=N THEN 1200
1100 MOVE (0,.434295*LOG(I)+J)
1110 IF I=10 THEN 1140
1120 DRAW (.05,.434295*LOG(I)+J)
1130 GOTO 1180
1140 FOR K=0 TO M STEP .05
1150 PLOT (K,.434295*LOG(I)+J)
1160 PENUP
1170 NEXT K
1180 MOVE (M,.434295*LOG(I)+J)
1190 DRAW (M-.05,.434295*LOG(I)+J)
1200 NEXT I
1210 NEXT J
1211 REM LABELS
1212 MOVE (1.3+G,-.12)
1213 PRINT #0;"CONCENTRATION",C$
1214 MOVE (-.1+G,-.12)
1215 PRINT #0;10^INT(.434295*LOG(C(Q)))
1216 MOVE (.9+G,-.12)
1217 PRINT #0;10^(INT(.434295*LOG(C(Q)))+1)
1220 MOVE (-.15,1.2)
1223 LDIR (90)
1225 PRINT #0;"LT50,HOURS"
1226 LDIR (0)
1227 MOVE (-.15,0)
1230 REM PLOTTING OF POINTS & Y SCALE LABELS
1232 IF T(1)>10 THEN 1239
1233 Q9=9
1235 PRINT #0;"1"
1236 MOVE (-.15,1)
1237 PRINT #0;"10"
1238 GOTO 1242
1239 PRINT #0;"10"
```

```
1240 MOVE (-.15,1)
1241 PRINT #0;"100"
1242 FOR I=1 TO Q
1243 A1=Z(I)
1244 IF Q9<>9 THEN 1247
1245 B1=D(I)
1246 GOTO 1250
1247 B1=D(I)-1
1250 GOSUB 1340
1260 NEXT I
1265 REM DRAWS & LABELS THRESHOLD
1266 X=.434295*LOG(-B/A)-INT(.434295*LOG(C(Q)))+G
1270 MOVE (X,N)
1272 LINE (6)
1280 DRAW (X,0)
1285 LINE (0)
1290 PENUP
1293 MOVE (.434295*LOG(-B/A)-INT(.434295*LOG(C(Q)))-.1+G,.5)
1294 LDIR (90)
1295 PRINT #0;"THRESHOLD";-B/A
1296 PENUP
1297 REM PLOTS LETHALITY CURVE
1298 X1=INT(43.4295*LOG(-B/A)+2)/100
1299 PRINT "PLOTTING CUTOFF";10^X1;"IF ERROR,INCREASE COEFF.IN 1298"
1300 FOR I=M+1 TO X1 STEP -.03
1302 Y8=-LOG(A+B/10^I)/K2
1303 IF Y8<0 THEN 1320
1307 IF Q9=9 THEN 1315
1310 PLOT (I-INT(.434295*LOG(C(Q)))+G,.434295*LOG(Y8)-1)
1311 GOTO 1320
1315 PLOT (I-INT(.434293*LOG(C(Q)))+G,.434295*LOG(Y8))
1320 NEXT I
1321 LDIR (0)
1322 MOVE (.5*M,.9*N)
1323 PRINT #0;T$,"THRESHOLD=";-B/A
1324 MOVE (.3*M,.8*N)
1325 PRINT #0;"LT50=-(1/";K2;")*LN(";A;"+";ABS(B);"/CW)"
1326 PRINT "FOR PLOTTER OUTPUT,TURN PLOTTER ON,RESET IF REQUIRED"
1327 INPUT "PLOTTER: YES (1), NO (2) ",Q8
1328 IF Q8=1 THEN 4000
1330 END
1340 MOVE (A1-.02+G,B1)
1350 DRAW (A1+.02+G,B1)
1360 MOVE (A1+G,B1+.03)
1370 DRAW (A1+G,B1-.03)
1380 RETURN
3000 FOR I=1 TO Q
3010 E(I)=EXP(-K2*T(I))
3020 NEXT I
3030 Z1,Z2,Z3,Z4,Z5=0
3040 FOR I=1 TO Q
3050 Z1=Z1+1/C(I)
3060 Z2=Z2+E(I)
3070 Z3=Z3+(1/C(I))^2
3080 Z4=Z4+(E(I))^2
3090 Z5=Z5+E(I)/C(I)
3100 NEXT I
3110 B=(Z5-Z1*Z2/Q)/(Z3-(Z1^2)/Q)
3120 A=(Z1*Z5-Z2*Z3)/((Z1^2)-Q*Z3)
3130 R=(Q*Z5-Z1*Z2)/SQR((Q*Z3-Z1^2)*(Q*Z4-Z2^2))
3140 PRINT "K2=";K2,"A=";A,"B=";B,"R=";R
3150 RETURN
4000 COMMAND "A S G"
4010 COMMAND "A D H#6"
4020 COMMAND "TR F"
4030 GOTO 1330
4500 PRINT "CALCULATION OF EXPONENTIAL DECREASE IN CW"
4510 INPUT "FOR PLOTTING PURPOSES, MAX.TIME ",T8
4520 INPUT "FOR PLOTTING PURPOSES,MAX.C ",C8
4530 INPUT "NUMBER OF POINTS ",N1
4540 LOCATE (100,200,30,100)
4550 FRAME
4560 SCALE (0,T8,0,C8)
4570 LAXES (-10,.1,0,0,5,1,2)
4580 Z1,Z2,Z3,Z4,Z5,Z6,N=0
4590 INPUT "CW ",C1
4600 INPUT "TIME ",T
4610 Z1=Z1+T
```

```
4620 Z2=Z2+T^2
4630 Z3=Z3+LOG(C1)
4640 Z4=Z4+(LOG(C1))^2
4650 Z5=Z5+T*LOG(C1)
4660 N=N+1
4670 GOSUB 4920
4680 IF N1=N THEN 4700
4690 GOTO 4590
4700 INPUT "ADDITIONAL POINTS TO PLOT ONLY:YES(1) NO(2) ",Q7
4710 IF Q7=2 THEN 4770
4720 INPUT "CW (TO END INPUT 0) ",C1
4730 IF C1=0 THEN 4770
4740 INPUT "TIME ",T
4750 GOSUB 4920
4760 GOTO 4720
4770 M=(Z5-Z1*Z3/N1)/(Z2-(Z1^2)/N1)
4780 C=(Z1*Z5-Z3*Z2)/(Z1^2-N1*Z2)
4790 R=(N1*Z5-Z1*Z3)/((N1*Z2-Z1^2)*(N1*Z4-Z3^2))^.5
4800 PRINT "SLOPE= ";M;" INTERCEPT = ";EXP(C);" CORREL.COEFF.= ";R
4810 M=-M
4820 GOSUB 4970
4830 PRINT "TO CALCULATE AVERAGE CONCN.,INPUT LT50"
4840 FOR I=1 TO Q
4850 INPUT "LT50 ",T(I)
4860 INPUT "NOMINAL CONCENTRATION ",C1
4865 IF T(I)>48 THEN 4872
4870 C(I)=-C1*EXP(C)*(EXP(-M*T(I))-1)/(M*T(I))
4871 GOTO 4880
4872 C(I)=-C1*EXP(C)*(EXP(-48*M)-1)/(48*M)
4873 C2=-C1*EXP(C)*(EXP(-M*(T(I)-48))-1)/(M*(T(I)-48))
4874 C(I)=.5*(C(I)+C2)
4880 PRINT "C";C(I),"LT50";T(I)
4890 NEXT I
4900 PRINT CHR$(141)
4910 GOTO 205
4920 MOVE (T-1,C1)
4930 DRAW (T+1,C1)
4940 MOVE (T,C1-.05*C1)
4950 DRAW (T,C1+.05*C1)
4960 RETURN
4970 PENUP
4980 FOR I=0 TO T8 STEP .4
4990 PLOT (I,EXP(C)*EXP(-M*I))
5000 NEXT I
5010 RETURN
```

```
RUN
LETCUR REV.2, APRIL 11,1980
COMPOUND NAME (MAX.18 CHARACTERS) CHLORDANE
NUMBER OF POINTS (MAX.20) 6
CHOOSE CONCENTRATION UNITS SO THAT VALUES ARE >1
CONCENTRATION UNITS (MG/L,UG/L),TYPE UNITS UG/L
A SUBROUTINE IS AVAILABLE FOR CALCULATION OF AVERAGE CONCENTRATIONS
SUBROUTINE: YES(1),NO(2) 2
INPUT CONCENTRATION AND LT50 IN DECREASING ORDER OF CONCENTRATIONS
POINT 1
CONCN. 124.5
LT50 22.4
POINT 2
CONCN. 114.6
LT50 23.6
POINT 3
CONCN. 102.3
LT50 34.1
POINT 4
CONCN. 39.2
LT50 70.9
POINT 5
CONCN. 30.8
LT50 82.7
POINT 6
CONCN. 24.9
LT50 92.9
K2 ESTIMATION BEGINS
K2 (RECOMMENDED K2=1) 1
K2= .5            A= 9.25522E-06                    B=-2.75228E-04
R=-.658591
K2= .25           A= 2.88161E-03                    B=-.0854552     R=-.712001
K2= .125          A= .0548735        B=-1.61882     R=-.801564
K2= .0625         A= .253955         B=-7.29165     R=-.891528
K2= .03125        A= .549304         B=-13.9329     R=-.944434
PRELIMINARY K2 ESTIMATE               .03125
------------------------------------------
K2 OPTIMIZATION BEGINS
K2= .015625       A= .782321         B=-15.0117     R=-.969214
K2= 7.8125E-03                       A= .90567      B=-11.3423      R=-.979985
-------------------------------------------------
INITIAL K2 RANGE                      7.8125E-03     .03125
INITIAL B RANGE                      -11.3423       -13.9329
-------------------------------------------
K2= .0195313     A= .719926          B=-15.3788     R=-.963331
K2= .0253906     A= .630736          B=-14.9669     R=-.954068
K2= .0136719     A= .81383           B=-14.536      R=-.972045
K2= .0136719     A= .81383           B=-14.536      R=-.972045
K2= .0166016     A= .766598          B=-15.1689     R=-.967769
K2= .0166016     A= .766598          B=-15.1689     R=-.967769
K2= .0180664     A= .743144          B=-15.3188     R=-.965568
K2= .0180664     A= .743144          B=-15.3188     R=-.965568
K2= .0187988     A= .7315            B=-15.3591     R=-.964455
K2= .0187988     A= .7315            B=-15.3591     R=-.964455
K2= .019165      A= .725704          B=-15.3715     R=-.963894
K2= .019165      A= .725704          B=-15.3715     R=-.963894
K2= .0193481     A= .722812          B=-15.3758     R=-.963614
K2= .0193481     A= .722812          B=-15.3758     R=-.963614
K2= .0194397     A= .721368          B=-15.3775     R=-.963472
$$$$$$$$$$$$$$$$$$$$$$$$$
CONVERGED. FINAL VALUES:
$$$$$$$$$$$$$$$$$$$$$$$$$
K2= .0194397     A= .721368          B=-15.3775
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$

                        LT50
 CONCENTRATION   --------------------
UG/L                MEASURED      CALCULATED
===============================================
 124.5              22.4          26.4617
 114.6              23.6          27.3881
 102.3              34.1          28.8212
 39.2               70.9          57.1735
 30.8               82.7          77.3996
 24.9               92.9          116.529
===============================================
$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$$
THRESHOLD IS      21.3171
```



Chlordane lethality curve

```
Lethality curve program for HP 29C. Documentation
--------------------------------------------------
Initial status of the calculator:
      Register 0 (R0)=16 (address of the first indirect storage register)
      Register 8 (R8)=16 (control value)
      Y-register =first LT50 value
      X-register =first CW value
--------------------------------------------------
01         LBL0            Data Entry Routine
      ==================================
02         1/x         Reciprocal of CW, currently in X-register
03         STO i       Store it in R16, the first indirect
                       storage register of HP 29C
04         ISZ         Increment indirect storage register by 1
05         x = y       Exchange the contents of the X- and Y-
                       registers. LT50 is now in the X-register
06         CHS         Convert LT50 to -LT50
07         STO i       Store -LT50 in R17
08         ISZ         Increment indirect storage register index
09         RTN         Return to LBL0. The calculator is ready to
                       accept another pair of values (LT50, CW).
                       AFTER COMPLETION OF TEST DATA ENTRY, THE
                       INDIRECT STORAGE REGISTERS CONTAIN:
                             R16    0.0333 (1/30)
                             R17   -0.7
                             R18    0.2     (1/5)
                             R19   -1.4
                             R20    2.0     (1/0.5)
                             R21   -3.3
                             R22    4.0     (1/0.25)
                             R23   -4.2
                       THE REGISTERS R24 - R29 ARE AVAILABLE FOR
                       ADDITIONAL VALUES (HP 29C CAN ACCEPT A TOTAL
                       OF 7 PAIRS)
10         LBL1        Statistical Summation Driver
           ==================================
11         CLEAR S     Clear the "statistical registers" R.0 - R.5.
                       The statistical registers of HP 29C store
                       sums used in statistical calculations:
                             R.0      n (number of entries)
                             R.1     S(x)
                             R.2     S(x^2)
                             R.3     S(y)
                             R.4     S(y^2)
                             R.5     S(x*y)
                       S (sigma on HP 29C) stands for sum
                       x,y are the current contents of X- and Y-
                       registers, respectively.
                       S + (sigma +) performs the addition (see 22)
12         DSZ         Decrease the indirect storage register index
                       to the value of the last entry in LBL0
13         GSB2        Call the LBL2
14         RTN         Return to LBL1. Calculation is completed.
15         LBL2        Statistical Summation Routine
           ==================================
```

```
16          RCL i       Recall last -LT50
17          RCL 9       Recall K2
18          *           Calculate -K2*LT50
19          e^x         Calculate EXP(-K2*LT50). THIS IS "Y" FOR
                        THE SUBSEQUENT REGRESSION(the dependent variable)
20          DSZ         Decrease indirect storage register index
21          RCL i       Recall CW corresponding to the current
                        LT50. THIS IS "X" FOR THE SUBSEQUENT
                        REGRESSION(the independent variable)
22          S +         (sigma +) Perform statistical summation
                        on the contents of X- and Y- registers
                        (CW and EXP(-K2*LT50), respectively)
23          DSZ         Decrease indirect storage register index
24          RCL 0       Recall the index into the X-register
25          RCL 8       Recall the "control value" 16 into the
                        X-register. This operation lifts the stack,
                        so that the indirect storage register index
                        is now in the Y-register
26          -           Calculate the difference:
                        Indirect storage register index - 16,
                        place it into the X-register
27          x>0         If the difference is positive, not all data
                        have been used as yet. EXECUTE STEP 28

                        If the difference is 0, all data have been
                        used. BYPASS STEP 28
28          GTO 2       Start again LBL2
29          GSB 3       Call LBL3
30          RTN         LBL2 is completed
                        WITH THE TEST DATA:
                            LT50         CW
                          --------    ---------
                            0.7          30
                            1.4          5
                            3.3          0.5
                            4.2          0.25
                          ----------------------
                         K2=0.4
                         ON COMPLETION OF THE STATISTICAL SUMMATIONS
                         (that is just before going to LBL3),
                         the contents of the statistical registers
                         are:
                        R.0       4
                        R.1       6.2333
                        R.2      20.0411
                        R.3       1.7805
                        R.4       1.0036
                        R.5       1.4192
                         THE REGRESSION EQUATION IS
                                   y = B*x + A
                        WHERE
                        B=[S(x*y)-S(x)*S(y)/n]/[S(x^2)^-(S(x))^2/n]
                        A=[S(x^2)*S(y)-S(x)*S(x*y)]/
                                /n*[S(x^2)-(S(x))^2/n]
                        AND THE CORRELATION COEFFICIENT r =
                        B*SQR[[S(x^2)-(S(x))^2/n]/
                              [S(y^2)-(S(y))^2/n]]

31          LBL3        Regression routine
=========================================================
32          RCL.1       S(x)
33          x^2         (S(x))^2
34          RCL.0       n
35          ./.         (S(x))^2/n
36          CHS         -(S(x))^2/n
37          RCL.2       S(x^2)
38          +           S(x^2)-(S(x))^2/n
39          STO 1       Denominator of A and B, store in R1
40          RCL.3       S(y)
41          x^2         (S(y))^2
42          RCL.0       n
43          ./.         (S(y))^2/n
44          CHS         -(S(y))^2/n
45          RCL.4       S(y^2)
46          +           S(y^2)-(S(y))^2/n
47          STO 2       Needed for r, store in R2
48          RCL.1       S(x)
49          RCL.3       S(y)
50          *           S(x)*S(y)
51          RCL.0       n
52          ./.         S(x)*S(y)/n
53          CHS         -S(x)*S(y)/n
54          RCL.5       S(x*y)
55          +           S(x*y)-S(x)*S(y)/n
56          STO 3       Numerator of B, store in R3
57          RCL.3       S(y)
58          RCL.2       S(x^2)
59          *           S(y)*S(x^2)
60          RCL.1       S(x)
61          RCL.5       S(x*y)
62          *           S(x)*S(x*y)
63          -           S(y)*S(x^2)-S(x)*S(x*y)
64          RCL.0       n
65          ./.         [S(y)*S(x^2)-S(x)*S(x*y)]/n
66          RCL 1       S(x^2)-(S(x))^2/n
67          STO./.3     Divide the contents of R3 by the contents
                        of the X-register and store the result
                        in R3; R3 now contains B
68          ./.         [S(y)*S(x^2)^S(x)*S(x*y)]/
                              /n*[S(x^2)-(S(x))^2/n]
                        THIS IS A
69          STO 4       store it in R4
```

```
70        RCL 2      S(y^2)-(S(y))^2/n
71        STO./.1    Divide the contents of R1 by the contents
                     of the X-register and store the result
                     in R1; R1 now contains
                     [S(x^2)-(S(x))^2/n]/[S(y^2)-(S(y))^2/n]
72        RCL 1      recall it into the R1 register
73        square root symbol   square root on X-register
74        RCL 3      B
75        *          X-register now contains the correlation
                     coefficient r
76        RTN        Regression calculations are completed
77        LBL9       Evaluation routine
          ===============================================
                     CALCULATOR STATUS:
                           R9 = K2
                           R4 = A
                           R3 = B
                           X-REGISTER = CW
78        1/x        1/CW
79        RCL 3      B
80        *          B/CW
81        RCL 4      A
82        +          A+B/CW
83        ln         ln(A+B/CW)
84        RCL 9      K2
85        ./.        ln(A+B/CW)/K2
86        CHS        X-register now contains calculated LT50
87        RTN        Calculation is completed
```

****

First step in the calculation is to estimate K2.
A good K2 value yields a correlation coefficient <-0.9 and the absolute value of B is usually at maximum.

KEY STROKES:
16 STO 0; STO 8; K2 STO 9 [initialization]
Key in LT50; ENTER; Key in CW; GSB 0 [repeat with each LT50,CW pair]
GSB1 [regression is calculated, correlation coefficient is displayed, A is stored in R4, B in R3.

TO REPEAT THE CALCULATION WITH A NEW VALUE OF K2:
Key in the new value of K2
STO 9;
Key in N+16 (N = number of data pairs times 2)
STO 0
GSB1

TO CALCULATE LT50 FROM CW:
Key in K2;
STO 9
Key in A;
STO 4
Key in B;
STO 3
[The above needed only when other than the current K2,A, and B values are used]

GSB 9

****

LBL3 is a general regression calculation and may be used for other purposes as well. Just make certain that the statistical registers contain the desired values and depress GSB3